AD-A064 880 ATHENS STATE COLL AL F/G 13/2

GAS COMPOSITIONAL ANALYSIS USING THE REAL TIME FIELD CONTAMINAT--ETC(U)

NOV 77 DAAK40-76-C-1027

UNCLASSIFIED NL

| OF |

AD A064880

END

DATE FILMED

4-79

DDC

AD-A064880

LEVEL II

1

FINAL REPORT

GAS COMPOSITIONAL ANALYSIS USING THE REAL TIME FIELD CONTAMINATION MONITOR

Contract DAAK40–76–C–1027

PRINCIPAL INVESTIGATOR: Dr. Gary L. Workman

November 1, 1977

DISTRIBUTION STATEMENT A
Approved for public release;
Distribution Unlimited

DDC
RECEIVED
26 FEB 1979
E

Prepared for:

U.S. ARMY MISSILE COMMAND AT REDSTONE ARSENAL, ALABAMA 35807

ATHENS STATE COLLEGE – Dept. of Science
ATHENS, ALABAMA 35611

79 02 09 022

FINAL REPORT

GAS COMPOSITIONAL ANALYSIS
USING THE
REAL TIME FIELD CONTAMINATION MONITOR

Contract DAAK40–76–C–1027

PRINCIPAL INVESTIGATOR: Dr. Gary L. Workman

November 1, 1977

Prepared for:

U. S. ARMY MISSILE COMMAND
AT
REDSTONE ARSENAL, ALABAMA 35807

ATHENS STATE COLLEGE
ATHENS, ALABAMA 35611

ACCESSION for
NTIS White Section X
DDC Buff Section
UNANNOUNCED
JUSTIFICATION
per Form 50 on file
BY
DISTRIBUTION/AVAILABILITY CODES
Dist. AVAIL. and/or SPECIAL
A

79 02 09 022

Abstract:

The development of additional hardware and software for the compositional analysis of gas mixtures using a gas chromatography - mass spectrometric technique has progressed to faster sampling rates and to a larger degree of automation. Hard copy in real time is also available now.

INTRODUCTION

The work performed under this contract has been primarily on extension of the previous work performed in the construction of a gas chromatograph - mass spectrometer system for use as a "Real Time Field Contamination Monitor." As the data acquisition system became more sophisticated, the mobility of the system began to decrease. In our earlier work on Contract DAAH01-75-C-0939, the contamination monitor incorporated a state-of-the-art programmable calculator, the Wang WCS-20 system, for the data acquisition and control functions. Described in this report is the further development of both hardware and software which implement these functions. Most of the design philosophy contained here has also been presented at a technical meeting last year.

Hardware Development

The use of the Wang system to control such things as sensitivity, start of scan, and duration of scan ideally can represent a true automation of the analysis. With the mass spectrometer setting at 10^{-11} amps a series of relays inside the UTI 100C become available for computer control of the range settings. Our circuitry to implement this control is given in Figure 1. A board layout of the Relay/Decoder Board is given in Figure 2. The decoder chosen to convert the Wang $\overline{\text{COB1}}$ - 4 binary signals to 16 control lines is the TTL logic 74154 decoder. As seen in Figure 1, all 8-bit data words from the Wang are now controlled by one of the 16 control lines through a 8212 buffered latch. The 74121 monostable vibrators are used to control pulse width for enabling the 8212's.

The arrangement as shown allows the user at least 128 lines which can operate as control functions. This includes all four $\overline{\text{COB1}}$ - 4 lines and all 8 data lines. It is easily seen that by expanding the 8 data lines through more 4 - to - 16 decoders, many more control functions would be accessible. We did not require this feature for our interfacing in this project.

The analog-to-digital convertor uses a Datel DAS-16 which takes approximately 25 microseconds to convert 8 bits. With the input scanner, up to 16 lines can be digitized, however, we did not require all those lines for this work. We did have some problems with the sequencer/control function of the DAS-16 which resulted in some down-times at inappropriate times. However, usually the A/D system operated very well.

The early work on this contract used the d.c. ramp from the mass spectrometer to drive the mass scale. Unfortunately, this scan approach

proved to be unreliable for mass calibration. For example, when taking an 0-100 amu scan in 4000 data points, the peak of a significant mass might differ by 10-15 points on the amu scale. This uncertainty in data points represented an uncertainty in mass value and since our mass values were calculated from the peak at the amu position, we never got the same answer on two different runs. This problem was alleviated by incorporating a digital ramp (or 2^{10} counter) as shown in Figure 3. The software increments the mass scale by one and this new count from the Wang data-out bus is latched out through the 8212's and counted by three 74193 up/down binary counters. The binary count is then converted by a 12 bit D/A DAC 85 (Burr - Brown) to provide the appropriate analog ramp. Using this technique, the mass calibration became more reliable. A board layout of the ramp driver board is given in Figure 4.

Also attempted during this work effort was the ability to record a real time with the particular spectrum being run. A real time clock with BCD outputs in months, days, hours, and minutes was being incorporated into the program. It is anticipated that this innovation will be part of the program in the future.

The chief drawback to the system above is the use of a silicone membrane for the gc-ms interface. We have experienced numerous problems in the membranes breaking due to overheating. Two electron multipliers have been ruined in this type of accident. Further recommendations might include a faster pumping system with no carrier gas.

Software Development

The software development has progressed in two directions under this particular contract. First, we have made use of the $GIO commands in order to speed up the data acquisition process. The $GIO commands operate much like machine language codes, thereby bypassing the Basic interpreter of the Wang system. This feature will be discussed further. The second direction in which we have gone in the software development is to real time plotting of the data using a Tektronix 4006-1 terminal with a hard copy unit. Enclosed as Figures 5, 6, and 7 are examples of this type of plot.

$GIO commands use a device code and hexadecimal characters to provide either input control or output control. For example rather than using the statements

```
SELECT INPUT 01
KEY IN A$, 50, 60
```

which represents an input code, we now use statements such as

```
$GIO/01D  (411D, Q$)
```

and since the BASIC interpreter is not needed we can take data (or output data) much faster. For example, our early programs would require 7 seconds to acquire 1000 data points; while our present system requires 2 seconds to acquire 4096 data points.

The $GIO commands are also used to transfer data to the 4006-1 terminal. We have increased the speed for the plotting of spectra; however, not to the times we had imagined. The principal drawback has been that to determine the graphical displacements on the CRT, division by 32 proceed very slowly. This technique needs to improve and will when more manpower becomes available.

The software is given in the following pages. "DATA 2" handles most of our mass spectrometry applications, while ANALOG 2 handles some plotting routines. TIME DECODE will allow the programmer to log the daily runs with a time specification.

Presentations and Personnel

The work presented in this report was also presented to the 28th Annual Southeast Regional Meeting of the American Chemical Society in Gatlinburg, Tennessee. Terrence Whitt presented the paper and subsequently received several offers for graduate school. He is now attending Murray State University and working on gc-ms applications there.

The primary program development has been performed by Mike Lenox. Mr. Lenox is still a student at Athens State College, but he will be graduating this year.

DATA ACQUISITION PROGRAM

```
10 REM ### SIMPLE PROGRAM FOR TAKING DATA ###
20 DEFFN'15
30 PRINT HEX(030A0A0A0A);"SAMPLING"
40 DIM A$(64,64)1
50 $GIO/23B (45014500,Q$)
60 $GIO/23A (CA70,Q$) A$()
70 INPUT "NAME",N$
80 DATA SAVE DC OPEN R 35,N$
90 DATA SAVE DC A$()
100 DATA SAVE DC END
110 PRINT N$;" HAS BEEN STORED"
120 STOP
130 FOR I=1 TO 64:FOR J=1 TO 64
140 PRINT 255-VAL(A$(I,J));
150 NEXT J:NEXT I
```

DATA 2 PROGRAM

```
10DEFFN'12"KEY S.F. '15 FOR INSTRUCTIONS"
20 DIM A$(64,64)1,R(100),P(100),Z$(100)9,B$(202,9)1
30 DEFFN'15
40 SELECT PRINT 005
50 PRINT HEX(030A0A)
60 PRINT TAB(20);"DEFFN'0 ---- TAKE DATA AND STORE"
70 PRINT TAB(20);"DEFFN'1 ---- LIST DATA FILES"
80 PRINT TAB(20);"DEFFN'2 ---- PICK PEAKS"
90 PRINT TAB(20);"DEFFN'3 ---- LIST DATA"
100 PRINT TAB(20);"DEFFN'4 ---- LIST PEAKS"
110 PRINT TAB(20);"DEFFN'5 ---- HARD COPY"
120 PRINT TAB(20);"DEFFN'6 ---- GRAPH PEAKS"
125 PRINT TAB(20);"DEFFN'7 ---- ANALOG GRAPH"
130 PRINT TAB(20);"DEFFN'15 --- RETURN TO INSTRUCTIONS"
140 STOP
145 DEFFN'7:LOAD DC F"ANALOG 3"
150 DEFFN'0 :REM          ####### TAKE DATA AND STORE #######
155 $GIO/23B(450C450041F9,Q$)
156 $GIO/23B(450B4500417F,Q$)
160 INPUT "SCAN LENGTH IN SECONDS",T
170 T=T*4.6248125
180 T1=INT(T/256):T2=T-256*T1
190 A$=HEX(0200030020)
200 BIN(STR(A$,2,2))=T1:BIN(STR(A$,4,2))=T2
210 $GIO/23B (A$,Q$)
220 $GIO/23B (1221 CB40 1200,Q$) A$()
230 INPUT "NAME",N$
240 DATA SAVE DC OPEN R 35,N$
250 DATA SAVE DC A$()
260 DATA SAVE DC END
270 PRINT N$;" HAS BEEN STORED"
280 STOP "KEY S.F. '15 FOR INSTRUCTIONS"
290 DEFFN'1:REM                ####### LIST DATA FILE #######
300 PRINT HEX(030A0A)
310 LIST DC R
320 STOP "KEY S.F. '15 FOR INSTRUCTIONS"
330 DEFFN'2:REM                        ####### PICK PEAKS #######
340 P,T=0
350 INPUT "WHICH DATA FILE",D$
360 INPUT "HOW MANY AMU (0-?)",A1
370 A1=INT(A1*64/400)
380 INPUT "CUTOFF POINT (0-255)",W
390 S=1 :REM          ####### SCALE CHOICE  DELETED #######
400 PRINT HEX(03)
410 ON S GOTO 420,430,440
```

DATA 2 PROGRAM (cont.)

```
420 C$="S 200":GOTO 450
430 C$="S 100":GOTO 450
440 C$="S 200"
450 DATA LOAD DC OPEN R D$
460 DATA LOAD DC A$()
470 DATA LOAD DC OPEN F C$
480 DATA LOAD DC B$()
490 FOR I=1 TO A1:FOR J=1 TO 64
500 PRINT HEX(01);INT((64*I+J)/10.24)-6,P
510 T=T+1
520 A=B:B=C:C=255-VAL(A$(I,J))
530 IF B[A THEN 590
540 IF B[C THEN 590
550 IF B=C THEN 590
560 IF B[W THEN 590
570 P=P+1:IF P]100 THEN 950
580 R(P)=B:P(P)=T
590 NEXT J:NEXT I
600 STOP "KEY S.F. '15 FOR INSTRUCTIONS"
610 DEFFN'6:REM            ####### TEKTRONIX SUB-ROUTINE #######
620 RESTORE
630 INPUT "DELTA X  (+ OR -)",D
640 INPUT "MULTIPLIER RANGE",M$
650 INPUT "PRESSURE",P$
660 FOR I=1 TO P
670 X=P(I)*1800/4096+120+D
680 M=M+1
690 FOR K=1 TO 2
700 IF K=1 THEN 710:Y=R(I)*2.55+115:GOTO 720
710 Y=115
720 X1=INT(X/32)+32:X2=((X/32)-INT(X/32))*32+64
730 Y1=INT(Y/32)+32:Y2=((Y/32)-INT(Y/32))*32+96
740 BIN(STR(Z$(M),K*4-3,1))=Y1:BIN(STR(Z$(M),K*4-2,1))=Y2:BIN(ST
R(Z$(M),K*4-1,1))=X1:BIN(STR(Z$(M),K*4,1))=X2
750 NEXT K
760 BIN(STR(Z$(M),9,1))=29
770 NEXT I
780 SELECT PRINT 01D
790 $GIO/01D (411B410C,Q$):$GIO/01D (411D,Q$)
800 FOR I=1 TO 100:NEXT I
810 DATA SAVE BT/41D,Z$()
```

DATA 2 PROGRAM (cont.)

```
820 DATA SAVE BT/41D,B$()
830 M=0
840 $GIO/01D(411F410D,Q$)
850 FOR I=1 TO 13:READ W$:PRINT W$:NEXT I
860 FOR I=1 TO 5:$GIO/01D(410D,Q$):NEXT I
870 PRINT TAB(8);"0";TAB(24);"50";TAB(39);"100";TAB(54);"150"
880 PRINT TAB(39);"M/E"
890 PRINT TAB(13);"FILE NAME--- ";D$
900 PRINT TAB(13);P;"PEAKS"
910 PRINT TAB(13);"CUTOFF POINT---";W
920 PRINT TAB(13);"MULTIPLIER RANGE--- ";M$
930 PRINT TAB(13);"PRESSURE--- ";P$
940 STOP "KEY S.F. '15 FOR INSTRUCTIONS"
950 W=W+10:P,T=0:PRINT W:GOTO 490
960 DEFFN'3:REM                     ####### LIST DATA #######
970 FOR I=1 TO 64:FOR J=1 TO 64
980 PRINT 255-VAL(A$(I,J));
990 NEXT J:NEXT I
1000 STOP "KEY S.F. '15 FOR INSTRUCTIONS"
1010 DEFFN'4:REM                    ####### PRINT PEAKS #######
1015 SELECT PRINT 005
1020 INPUT "ON TYPEWRITER",L$
1030 SELECT P3
1040 IF STR(L$,1,1)[]"Y" THEN 1070
1050 SELECT P0
1060 SELECT PRINT 211
1070 PRINT "     AMU     RELATIVE INTENSITY"
1080 FOR I=1 TO P:PRINTUSING 1110,INT((P(I)/10.04)+.5) ,R(I):NEX
T I
1090 SELECT P0
1100 STOP "KEY S.F. '15 FOR INSTRUCTIONS"
1110%    ###         ###
1120 DATA "R","E","L"," ","I","N","T","E","N","S","I","T","Y"
1130 DEFFN'5:REM                 ####### HARD COPY #######
1140 $GIO/01D (411B4117,Q$)
1150 STOP "KEY S.F. '15 FOR INSTRUCTIONS"
```

ANALOG 2 PROGRAM

```
10 SELECT PRINT 005
20 PRINT HEX(030A0A)
30 $GIO/01D(411B410C,Q$)
40 SELECT PRINT 01D
50 DIM A$(64,64)1,Z$5
60 INPUT "WHICH DATA FILE",F$
70 INPUT "AMU RANGE----BEGIN",A1:INPUT "                END",A2
80 $GIO/01D(411D,Q$):X=700:Y=30:GOSUB '50:$GIO/01D(411F,Q$)
90 PRINT "FILE NAME IS ";F$
100 DATA LOAD DC OPEN R F$
110 DATA LOAD DC A$()
120 $GIO/01D(411D,Q$)
130 X=99:Y=114:GOSUB '50:X=1000:Y=114:GOSUB '50
140 X=1000:Y=775:GOSUB '50:X=99:Y=775:GOSUB '50
150 X=99:Y=114:GOSUB '50:$GIO/01D(411D,Q$)
160 X=96:Y=111:GOSUB '50:X=1003:Y=111:GOSUB '50
170 X=1003:Y=778:GOSUB '50:X=96:Y=778:GOSUB '50
180 X=96:Y=111:GOSUB '50:$GIO/01D(411D,Q$)
190 $GIO/01D(411D,Q$)
200 I=INT(A1*10.24/64)+1:J=INT((A1*10.24)-(64*INT(A1*10.24/64)))

220 X=100
230 Q=900/(10.24*(A2-A1))
240 J=J+1:IF J[65 THEN 250:J=1:I=I+1
250 X=X+Q
260 IF X]1000 THEN 310
270 SELECT PRINT 01D
280 Y=2.55*(255-VAL(A$(I,J)))+115
290 GOSUB '50
300 GOTO 240
310 $GIO/01D(411D,Q$)
320 X=75:Y=65:GOSUB '50:$GIO/01D(411F,Q$):PRINT A1:$GIO/01D(411D
,Q$)
330 $GIO/01D(411D,Q$)
340 IF A2-A1]100 THEN 350:S=1:GOTO 380
350 IF A2-A1]200 THEN 360:S=2:GOTO 380
360 IF A2-A1]300 THEN 370:S=3:GOTO 380
370 S=4
380 X=10:Y=35:GOSUB '50:$GIO/01D(411F,Q$):PRINT "INCREMENT=";S;"
AMU":$GIO/01D(411D,Q$)
390 X=100
400 FOR I=A1 TO A2 STEP S
410 Y=110
420 GOSUB '50
430 IF Y[110 THEN 480
```

ANALOG 2 PROGRAM (cont.)

```
440 O=O+1:IF O[11 THEN 450:O=1:Y=85:GOSUB '50:GOTO 460
450 Y=100:GOSUB '50
460 $GIO/01D(411D,Q$)
470 X=X+(900/(A2-A1))*S
480 NEXT I
490 X=975:Y=65:GOSUB '50:$GIO/01D(411F,Q$):PRINT A2:$GIO/01D(411
D,Q$)
500 STOP "KEY '15 TO RETURN TO INSTRUCTIONS"
510 DEFFN'50
520 X1=INT(X/32)+32:X2=((X/32)-INT(X/32))*32+64
530 Y1=INT(Y/32)+32:Y2=((Y/32)-INT(Y/32))*32+96
540 BIN(A$)=Y1:BIN(B$)=Y2:BIN(C$)=X1:BIN(D$)=X2
550 SELECT PRINT 01D
560 PRINT A$;B$;C$;D$;
570 RETURN
580 DEFFN'15:LOAD DC F"DATA 2"
```

TIME DECODE PROGRAM

```
10 REM    ####### TIME DECODE PROGRAM ########
20 REM ## TIME COMES IN IN 3 BYTES AND IS CONVERTED TO ##
      ## MONTH,DAY,HOUR AND MINUTES                    ##
30 DIM T$(3)1
40 $GIO/23A (C370,Q$) T$
50 T1$,T2$,T3$,T4$=T$(1):U1$,U2$,U3$=T$(2):V1$,V2$,V3$=T$(3)
60 AND (T1$,80):AND (T2$,78):AND (T3$,06):AND (T4$,01)
70 AND (U1$,E0):AND (U2$,18):AND (U3$,07)
80 AND (V1$,80):AND (V2$,70):AND (V3$,0F)
90 M=VAL(T1$)/128*10+VAL(T2$)/8
100 D=VAL(T3$)/2*10+8*VAL(T4$)+VAL(U1$)/32
110 H=VAL(U2$)/8*10+VAL(U3$)*2+VAL(V1$)/128
120 M1=VAL(V2$)/16*10+VAL(V3$)
130 PRINTUSING 140,M,D,H,":",M1
140 %  THE DATE IS  ##/##  AND THE TIME IS  ## # ##
```



Figure 1. Relay/Decoder Board Schematic



R = 10kΩ C = .1μF

Figure 2. Board Layout of Relay/Decoder Board



Figure 3. Ramp Drive Board Schematic



R1 - 5.6KΩ
R2 - 1.8MΩ
R3 - 3.9MΩ
R4, R5, R6 - 10KΩ

C1 - .0022μF
C2, C3, C4 - .1μF

P1, P2 - 50KΩ Potentiometer

Figure 4. Ramp Drive Board Layout



Figure 5. Typical bar output for mass spectrometer



Figure 6. Typical spectrum of dirty chamber.

DAT80403

M/E	PEAK HEIGHT
28.0	10.00
49.3	10.00
9.4	9.49
14.0	8.11
26.5	8.11
8.4	7.76
13.3	7.52
70.7	7.52
7.8	6.82
29.7	5.60



Figure 7. Example of peak printout and display.